HW1 Exercises from Rinzel lecture #1, Sept 14, 2005.
Due Sept 28, 2005.

Choose 2 out of the following 4 exercises. See following pages for the model's eqns and description of Euler's method for numerical integration.

1. Consider HH without I_K (ie, g_K =0). Show that with adjustment in g_{Na} (and maybe g_{leak}) the HH model is still excitable and generates an action potential. (Do it with $m=m_\infty(V)$.) Study this 2 variable (V-h) model in the phase plane: nullclines, stability of rest state, trajectories, etc. Then consider a range of I_{app} to see if you get repetitive firing. Compute the freq vs I_{app} relation; study in the phase plane. Do analysis to see that the rest point must be on the middle branch to get a limit cycle.

2. Convert the HH model into "phasic mode". By "phasic" I mean that the neuron does not fire repetitively for any I_{app} values – only 1 to a few spikes and then it returns to rest. Many neurons in the auditory system behave phasically. Do this by, say, sliding some channel-gating dynamics along the V-axis (probably just for I_K) . [If you slide $x_\infty(V)$, you must also slide $\tau_x(V)$.] If it can be done using h=1-n and $m=m_\infty(V)$ then do the phase plane analysis.

3. Consider the FitzHugh-Nagumo model and describe its repetitive firing properties in terms of Hopf bifurcation theory:
$$v' = -f(v) - w + I_{app}$$
$$w' = \varepsilon\,(v - \gamma\, w)$$
where $f(v)=v(v-a)(v-1)$; $0 \le a \le 1$; $\varepsilon,\gamma > 0$.
   a. Show that the rest state (v_R, w_R) is unique if γ is small enough
   $$\gamma < 3/(a^2-a+1).$$
   b. Find analytically the parameter conditions such that Hopf bifurcations occur for some critical current values $I_{app}=I_1, I_2$.
   Answer: $3\varepsilon\gamma < a^2-a+1$.
   Find expressions for I_1, I_2 in terms of ε, γ, a. (Hint: first use v_R as your control parameter and then later compute I_1, I_2 .) Plot $I_{1,2}$ versus ε (same axes). Interpret the results in terms of the repetitive firing regime, ε as "temperature", and the Hopf-predicted frequency.
   c. With numerical simulations (or AUTO in XPP) compare the repetitive firing properties for ε=0.07 and ε=0.02 with a=0.1 and γ=1 – compute frequency vs. I_{app}, and amplitude (v_{max}, v_{min} vs. I_{app}).

4. Explore the Morris-Lecar model. For parameter values in the Handout, obtain, plot and discuss the time courses for I=20, 40, 60, 120. Use numerical integration for $0<t<900$, starting from the rest state (for I=0). (I used Runge-Kutta with Δt=0.2, but you could use Euler – maybe with a smaller step size.) Construct and describe features of the phase plane, w vs v, for each of these cases: nullclines, singular points, stability, sample trajectories. For which values of I is the system

excitable, oscillatory, in nerve block, etc. Consult the Borisyuk/Rinzel Chapt, Fig 14, if you wish.

HH and WC models. Sept 14, 2005
Euler method for numerical integration.

HH Equations.
XPP code.

```
#hh4jr.ode
p VNA=115,VK=-12,VL=10.5989,GNABAR=120,GKBAR=36,GL=.3
p I0=0,IP=0,PON=0,POFF=0,TEMP=6.3
AM(v)= PHI*0.1*(25.-v)/(EXP(0.1*(25.-v))-1.)
BM(v)=PHI*4.0*EXP(-v/18.)
AH(v)=PHI*0.07*EXP(-v/20.)
BH(v)=PHI*1.0/(EXP(0.1*(30-v))+1.0)
BN(v)=PHI*.125*EXP(-v/80.)
AN(v)=PHI*0.01*(10.-v)/(EXP(0.1*(10.-v))-1.0)
IAPP(t)=I0+heav(POFF-t)*heav(t-PON)*ip
PHI= 3**((TEMP-6.3)/10.)
#minf(v)=am(v)/(am(v)+bm(v))
#iionminf(v,h,n)=GNABAR*MINF(V)**3*H*(V-VNA)+GKBAR*N**4*(V-
VK)+GL*(V-VL)
dv/dt= IAPP(T)-GNABAR*M**3*H*(V-VNA)-GKBAR*N**4*(V-VK)-GL*(V-VL)
dm/dt= AM(V)-(AM(V)+BM(V))*M
dh/dt= AH(V)-(AH(V)+BH(V))*H
dn/dt= AN(V)-(AN(V)+BN(V))*N
init v=20.001 m=.05932 h=.59612 n=.31768
@ METH=qualrk,TOL=.001,BOUND=100000,xlo=0,xhi=20,ylo=-20,yhi=140
done
```

WC Equations.
XPP code.

```
init u=.0426,v=.0834
u'=-u+f(aee*u-aie*v+Je+stim(t))
v'=(-v+f(aei*u-aii*v+Ji))/tau
par aee=10,aie=9,Je=-3
par aei=20,aii=3,Ji=-3,tau=5
stim(t)=s0+s1*if(t<tdone)then(t/tdone)else(0)
par s0=0,tdone=1,s1=0
f(u)=1/(1+exp(-u))
@ xp=u,yp=v,xlo=-.1,ylo=-.1,xhi=1.1,yhi=1.1,total=50
done
```

ML Equations

XPP code.

```
# Morris-Lecar model
dv/dt=(i+gl*(vl-v)+gk*w*(vk-v)+gca*minf(v)*(vca-v))/c
dw/dt=lamw(v)*(winf(v)-w)
minf(v)=.5*(1+tanh((v-v1)/v2))
winf(v)=.5*(1+tanh((v-v3)/v4))
lamw(v)=phi*cosh((v-v3)/(2*v4))
param vk=-84,vl=-60,vca=120
param i=0,gk=8,gl=2, gca=4, c=20
param v1=-1.2,v2=18,v3=12,v4=17,phi=.066666667
# for type II dynamics, use v3=2,v4=30,phi=.04
# for type I dynamics, use v3=12,v4=17,phi=.06666667
v(0)=-60.899
w(0)=.0148
# track some currents
aux Ica=gca*minf(V)*(V-Vca)
aux Ik=gk*w*(V-Vk)
done
```

Euler's Method.

For numerical simulation you could implement the forward Euler scheme. Suppose that $t_n = n\,dt$ where dt is a small time step, say dt=0.02 ms for HH (or smaller if necessary) .

Then if x_n is the approximation to $x(t_n)$ the Euler recipe for "integrating " (getting the approximate time course, recursively) the ode:

$$dx/dt = f(x), \quad x_0=x(0)$$

is:

$$x_{n+1} = x_n + dt \; f(x_n)$$

where x_o is the starting (initial) value of x at t=0. There you go, just march forward in time (ie, n) repeating this recipe successively. Save $(t_{n,}, x_n)$ as an array and plot it, x_n vs t_n, to see the time course. Do at least one run where you reduce the dt by ½ to see that you get good agreement in the time courses. This is an accuracy test.